
Matrixformler 

Matrixformler gør egentlig det samme som datatabeller (se artiklen om disse), nemlig regner på 

flere celler på en gang. De kan ofte træde i stedet for tabeller (som dog er hurtigere og nemmere at 

lave), men der er også situationer, hvor tabellerne ikke kan bruges og det er nødvendigt at anvende 

matrixformler i stedet. En del af løsningseksempler i på hjemmesiden er netop baseret på, at der 

anvendes matrixformler.  

 

I lighed med tabelformlerne, indtaster man selv sine matrixformler, men disse afsluttes ikke som 

sædvanligt med Enter, klik på  eller ved at flytte markøren til en anden celle. En matrixformel 

kan kun oprettes ved at trykke tastekombinationen Ctrl+Skift+Enter. Hvis vi vender tilbage til det 

eksempel jeg også brugte under tabeller, skulle vi beregne den månedige ydelse på et lån ved va-

rienrende rentesatser. Disse var placeret i celle A5 til celle A40. I A2 stod løbetiden og i A3 lånebe-

løbet. 

 

Vi kan nu oprette vores formler på to måder. I den første opretter vi den i en celle ad gangen. I celle 

B5 skrives formlen =YDELSE(A5:A40/12;A$2;A$3) og der afsluttes med Ctrl+Skift+Enter. Da der 

er tale om en matrixformel, vil den regne på den celle i matrixen, som står ud for formlen, altså A5. 

Ser vi på formlen, kan vi se, at den ligesom i tabelformlerne er blevet omkranset af tuborgparente-

ser. Som i tabelformler kan vi ikke skrive tuborgparentserne selv, de skal oprettes med 

Ctrl+Skift+Enter.  De absolutte cellereferencer på de to andre argumenter er nødvendige, da form-

len nu skal kopieres nedad til B40. Så er beregningen færdig.  

 

Som alternativ til denne metode, kan vi nøjes med at indtaste formlen uden at skulle kopiere, og 

uden at have brug for absolutte referencer. I dette tilfælde starter vi med at markere alle cellerne fra 

B5 til B40 inden vi begynder at taste formlen, som ellers tastes på fuldstændigt samme måde. Når 

der nu trykkes Ctrl+Skift+Enter, indsættes der formler i alle de markerede celler på én gang.  

 

Sidstnævnte metode er hurtigere end den først anviste. Til gengæld vil den første metode give 36 

enkelte matrixformler, mens den sidste vil give en samlet formel fordelt over alle cellerne. Resul-

tatmæssigt er der ingen forskel, men som med datatabellerne er det ikke muligt at ændre i en del af 

en matrix; kun hele matrixen på én gang kan ændres eller slettes. Derfor kan det af og til være en 

fordel at have separate matrixformler i hver celle, mens det i andre sammenhænge ikke betyder no-

get.  

 

Også med to variable kan man bruge en matrixformel. I så fald anvender man "bare" to matricer i 

sin formel. Med rentesatser i A5 til A40 og løbetider i B4 til H4, kan man markere området B5 til 

H40 og så indtaste denne formel =YDELSE(A5:A40/12;B4:H4;A3) og så afslutte med 

Ctrl+Skift+Enter. Så oprettes nedenstående regneark.  

Beregningerne giver nøjagtigt de samme resultater som i tabellerne, men som det kan ses i formel-

linjen i figuren, indgår de variable celler nu direkte i formlen i form af matricerne, ikke indirekte i 

form at inputcellerne som i tabelvarianten.  

 

Matrixformler skal/kan ikke kun bruges ved tabellignende opstillinger. Der er også en hel del en-

keltsituationer, hvor kun en matrixformel giver det rigtige resultat. 

 

Forestil dig dette regneark: 

 

Mærke Farve Status Pris

Volvo Rød Solgt 300000

Renault Gul 298000

Ford Grå Solgt 149000

Renault Sølv 268000

WV Rød Solgt 340000

Citroen Sølv Solgt 137000

Ford Sort 342000

Opel Sølv Solgt 153000

Toyota Grøn 234000

Ib's brugte biler

Vi vil nu gerne finde den næstdyreste bil, af dem, som er solgt. Funktionen STØRSTE() kan bruges 

til at finde den næststørste værdi i en matrix. Vil vi fx finde den næsthøjeste pris i kolonnen med 

priser, her D, kan denne formel bruges: =STØRSTE(D3:D11;2). Første argument, er området, der 

skal undersøges og andet argument er den ønskede værdis ”nummer”. 2 angiver således at det skal 

være den næststørste, 3 den tredjestørste og så fremdeles. Dette er imidlertid ikke nok her, da der 

samtidig er den betingelse, at bilen skal være solgt. Havde det været en celle, der skulle undersøges 

for en betingelse, kunne HVIS() have gjort det, eller hvis vi skulle have talt eller summeret. Ckunne 

man have brugt SUMPRODUKT, men her må vi gå en anden vej.  En formel som 

=HVIS(D3:D11;… vil ikke umiddelbart virke, da HVIS() kun kan regne på en celle ad gangen. 

Men som en matrixformel går det. Vi kan nu opbygge følgende samlede formel: 

 

=STØRSTE(HVIS(C3:C11="solgt";D3:D11);2) 

 

Hvis vi bare trykker Enter finder den den største værdi, 340.000, men indtaster vi den som en ma-

trixformel, finder den, som ønsket 300.000, som er den næsthøjeste pris, vi har solgt en bil til. For-

klaringen er, at matrixformlen får Excel til at undersøge hver af cellerne C3:C11 og se, om der står 

”Solgt” i dem. Gør der det, tager den værdien fra den tilsvarende række i D-kolonnen. STØRSTE() 

sørger så for, at det er den næststørste af disse værdier, som skrives som resultat i formelcellen. 

 

Under SmåTips kan man findes yderligere nogle eksempler. Her vil jeg bare illustrere med et enkelt 

af disse. 

 

Dato Aflæsning

23-02-2010 2144

24-02-2010 2453

25-02-2010 2789

26-02-2010 3011

27-02-2010 3244

28-02-2010 3456

01-03-2010 3700

02-03-2010 3990

03-03-2010 4287

04-03-2010 4567

05-03-2010 4812

06-03-2010 5132

07-03-2010 5341

08-03-2010 5788

09-03-2010 6001

10-03-2010 6456

11-03-2010 6745

12-03-2010 7122

13-03-2010 7456

14-03-2010 7567  
 

I et regneark har jeg følgende måleraflæsninger. Ved at indsætte formlen B3-B2 i celle C3 og kopi-

ere den nedad, kan jeg beregne hver dags forbrug. Hvis jeg vil finde det største forbrug i perioden, 

kan jeg lave en =MAKS(C3:C21). Hvis jeg imidlertid bare vil finde det største forbrug, uden at 

beregne de enkelte dages forbrug, kan jeg bruge =MAKS(B3:B21-B2:B20) indtastet som matrix-

formel. Hver celle i de to områder B3:B21 og B2:B20 vil blive trukket fra hinanden to og to, B3-

.B2, B4-B3 osv., og MAKS() finder så det største af disse tal. 

 

 

Matrixkonstanter 

Hvis man ikke vil indtaste alle sine konstante værdier i hver sine celler i regnearket, kan man i ste-

det anvende såkaldte Matrixkonstanter. Start med at markere, de celler, som konstanterne ellers 

skulle stå i. Opret dernæst matrixkonstanten, som skal stå i cellerne og tryk til sidst 

Ctrl+Skift+Enter.  

 

Selve matrixkonstanten indtastes som fx ={1;2;3}. Her er altså en undtagelse, hvor vi selv skal taste 

tuborg-parenteserne. Har jeg markeret tre celler, fx A1, B1 og C1, vil der nu i hver af cellerne stå 

{={1;2;3}} hvoraf de yderste tuborg-parenteser kommer fra Ctrl+Skift+Enter. I hver af de tre celler 

vises dog kun én af de valgte værdier. Ovenstående eksempel bruges hvis konstanten er vandret, 

altså flere kolonner i samme række. Skal vi i stedet have flere rækker i samme kolonne, skal man 

indtaste sin matrixkonstant som ={1\2\3}, altså med blackslash som skilletegn, ikke komma, som 

der står i hjælpen til Excel. 

 

Men hvorfor nu have det besvær, når vi bare kunne have tastet et enkelt tal i hver celle? Af "sikker-

hedsgrunde".  Taster vi et tal i hver celle, kan en bruger nemt komme til at overskrive et af disse tal. 

Taster vi som en matrixkonstant, kan dette ikke ske, da det jo, som beskrevet ovenfor, ikke er mu-

ligt at ændre en del af en matrix. Det er derfor ikke muligt ved en fejl at ændre de konstante værdi-

er.  

 

Selve matrixkonstanten kan indeholde tal, både heltal og decimaltal. Desuden kan den indeholde 

tekster, som, ligesom de fleste andre steder i Excel, skal skrives i anførselstegn, fx "Jan". Endelig 

kan den indeholde logiske værdier som SAND og FALSK samt fejlværdier. Derimod må den ikke 

indeholde celleadresser, formler, samt $-tegn, %-tegn
1
 eller parenteser. Her et eksempel på en ma-

trixkonstant, der inder både tekst, tal og logiske værdier. Læg mærke til, at selv om de logiske vær-

dier indtastes som tekst, skal der ikke anførselstegn omkring, da de så netop vil blive opfattet som 

tekst, ikke som den logiske værdi. 

 

{={1;2;3;"Jan";"Poul";"Niels";SAND;FALSK;SAND}} 

 

 

 

 

 

 

 

 

                                                 
1
  I stedet for at bruge %-tegn, kan man omregne procentsatserne til decimaler. 


